NAG Library Routine Document

F08KPF (ZGESVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08KPF (ZGESVD) computes the singular value decomposition (SVD) of a complex m by n matrix A, optionally computing the left and/or right singular vectors.

2 Specification

```
SUBROUTINE F08KPF (JOBU, JOBVT, M, N, A, LDA, S, U, LDU, VT, LDVT, WORK,        &
                   LWORK, RWORK, INFO)

INTEGER                 M, N, LDA, LDU, LDVT, LWORK, INFO
REAL (KIND=nag_wp)      S(*), RWORK(*)
COMPLEX (KIND=nag_wp) A(LDA,*), U(LDU,*), VT(LDVT,*), WORK(max(1,LWORK))
CHARACTER(1)            JOBU, JOBVT
```

The routine may be called by its LAPACK name ***zgesvd***.

3 Description

The SVD is written as

$$A = U\Sigma V^{\mathrm{H}},$$

where Σ is an m by n matrix which is zero except for its $\min(m,n)$ diagonal elements, U is an m by m unitary matrix, and V is an n by n unitary matrix. The diagonal elements of Σ are the singular values of A; they are real and non-negative, and are returned in descending order. The first $\min(m,n)$ columns of U and V are the left and right singular vectors of A.

Note that the routine returns V^{H}, not V.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBU – CHARACTER(1) *Input*

On entry: specifies options for computing all or part of the matrix U.

JOBU = 'A'
All m columns of U are returned in array U.

JOBU = 'S'
The first $\min(m,n)$ columns of U (the left singular vectors) are returned in the array U.

JOBU = 'O'
The first $\min(m,n)$ columns of U (the left singular vectors) are overwritten on the array A.

JOBU = 'N'
No columns of U (no left singular vectors) are computed.

Constraint: JOBU = 'A', 'S', 'O' or 'N'.

2: JOBVT – CHARACTER(1) *Input*

On entry: specifies options for computing all or part of the matrix V^{H}.

JOBVT = 'A'
All n rows of V^{H} are returned in the array VT.

JOBVT = 'S'
The first $\min(m,n)$ rows of V^{H} (the right singular vectors) are returned in the array VT.

JOBVT = 'O'
The first $\min(m,n)$ rows of V^{H} (the right singular vectors) are overwritten on the array A.

JOBVT = 'N'
No rows of V^{H} (no right singular vectors) are computed.

Constraints:

JOBVT = 'A', 'S', 'O' or 'N';
JOBVT and JOBU cannot both be 'O'.

3: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $\mathrm{M} \geq 0$.

4: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

5: A(LDA,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1,\mathrm{N})$.

On entry: the m by n matrix A.

On exit: if JOBU = 'O', A is overwritten with the first $\min(m,n)$ columns of U (the left singular vectors, stored column-wise).

If JOBVT = 'O', A is overwritten with the first $\min(m,n)$ rows of V^{H} (the right singular vectors, stored row-wise).

If JOBU $\neq$ 'O' and JOBVT $\neq$ 'O', the contents of A are destroyed.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08KPF (ZGESVD) is called.

Constraint: $\mathrm{LDA} \geq \max(1,\mathrm{M})$.

7: S($*$) – REAL (KIND=nag_wp) array *Output*

Note: the dimension of the array S must be at least $\max(1,\min(\mathrm{M},\mathrm{N}))$.

On exit: the singular values of A, sorted so that $\mathrm{S}(i) \geq \mathrm{S}(i+1)$.

8: U(LDU,$*$) – COMPLEX (KIND=nag_wp) array *Output*

Note: the second dimension of the array U must be at least $\max(1, \mathrm{M})$ if JOBU = 'A', $\max(1, \min(\mathrm{M}, \mathrm{N}))$ if JOBU = 'S', and at least 1 otherwise.

On exit: if JOBU = 'A', U contains the m by m unitary matrix U.

If JOBU = 'S', U contains the first $\min(m, n)$ columns of U (the left singular vectors, stored column-wise).

If JOBU = 'N' or 'O', U is not referenced.

9: LDU – INTEGER *Input*

On entry: the first dimension of the array U as declared in the (sub)program from which F08KPF (ZGESVD) is called.

Constraints:

if JOBU = 'A' or 'S', LDU $\geq \max(1, \mathrm{M})$;
otherwise LDU ≥ 1.

10: VT(LDVT,$*$) – COMPLEX (KIND=nag_wp) array *Output*

Note: the second dimension of the array VT must be at least $\max(1, \mathrm{N})$ if JOBVT = 'A' or 'S', and at least 1 otherwise.

On exit: if JOBVT = 'A', VT contains the n by n unitary matrix V^{H}.

If JOBVT = 'S', VT contains the first $\min(m, n)$ rows of V^{H} (the right singular vectors, stored row-wise).

If JOBVT = 'N' or 'O', VT is not referenced.

11: LDVT – INTEGER *Input*

On entry: the first dimension of the array VT as declared in the (sub)program from which F08KPF (ZGESVD) is called.

Constraints:

if JOBVT = 'A', LDVT $\geq \max(1, \mathrm{N})$;
if JOBVT = 'S', LDVT $\geq \max(1, \min(\mathrm{M}, \mathrm{N}))$;
otherwise LDVT ≥ 1.

12: WORK($\max(1, \mathrm{LWORK})$) – COMPLEX (KIND=nag_wp) array *Workspace*

On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.

13: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08KPF (ZGESVD) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK should generally be larger. Consider increasing LWORK by at least $nb \times \min(\mathrm{M}, \mathrm{N})$, where nb is the optimal ***block size***.

Constraint: LWORK $\geq \max(1, 2 \times \min(\mathrm{M}, \mathrm{N}) + \max(\mathrm{M}, \mathrm{N}))$.

14: RWORK($*$) – REAL (KIND=nag_wp) array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, 5 \times \min(\mathrm{M}, \mathrm{N}))$.

On exit: if INFO > 0, RWORK$(1 : \min(\mathrm{M}, \mathrm{N}) - 1)$ contains the unconverged superdiagonal elements of an upper bidiagonal matrix B whose diagonal is in S (not necessarily sorted). B satisfies $A = UBV^{\mathrm{H}}$, so it has the same singular values as A, and singular vectors related by U and V^{H}.

15: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If F08KPF (ZGESVD) did not converge, INFO specifies how many superdiagonals of an intermediate bidiagonal form did not converge to zero. See the description of RWORK above for details.

7 Accuracy

The computed singular value decomposition is nearly the exact singular value decomposition for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. In addition, the computed singular vectors are nearly orthogonal to working precision. See Section 4.9 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating point operations is approximately proportional to mn^2 when $m > n$ and m^2n otherwise.

The singular values are returned in descending order.

The real analogue of this routine is F08KBF (DGESVD).

9 Example

This example finds the singular values and left and right singular vectors of the 6 by 4 matrix

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i & -0.05 + 0.41i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i & -0.81 + 0.56i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i & -1.11 + 0.60i \\ -0.37 + 0.38i & 0.19 - 0.54i & -0.98 - 0.36i & 0.22 - 0.20i \\ 0.83 + 0.51i & 0.20 + 0.01i & -0.17 - 0.46i & 1.47 + 1.59i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i & 0.26 + 0.26i \end{pmatrix},$$

together with approximate error bounds for the computed singular values and vectors.

The example program for F08KRF (ZGESDD) illustrates finding a singular value decomposition for the case $m \leq n$.

9.1 Program Text

```
    Program f08kpfe

!     F08KPF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, zgesvd
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                    :: nb = 64, nin = 5, nout = 6,    &
                                               prerr = 0
!     .. Local Scalars ..
      Integer                               :: i, info, lda, ldu, ldvt,       &
                                               lwork, m, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable    :: a(:,:), a_copy(:,:), b(:),     &
                                               u(:,:), vt(:,:), work(:)
      Complex (Kind=nag_wp)                 :: dummy(1,1)
      Real (Kind=nag_wp), Allocatable       :: rwork(:), s(:)
!     .. Intrinsic Procedures ..
      Intrinsic                             :: max, min, nint, real
!     .. Executable Statements ..
      Continue
      Write (nout,*) 'F08KPF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      ldu = m
      ldvt = n
      Allocate (a(lda,n),a_copy(m,n),s(n),u(ldu,m),vt(ldvt,n),b(m),rwork(5*n))

!     Read the m by n matrix A from data file
      Read (nin,*)(a(i,1:n),i=1,m)

!     Read the right hand side of the linear system
      Read (nin,*) b(1:m)

      a_copy(1:m,1:n) = a(1:m,1:n)

!     Use routine workspace query to get optimal workspace.
      lwork = -1
!     The NAG name equivalent of dgesvd is f08kpf
      Call zgesvd('A','S',m,n,a,lda,s,u,ldu,vt,ldvt,dummy,lwork,rwork,info)

!     Make sure that there is enough workspace for blocksize nb.
      lwork = max(m+3*n+nb*(m+n),nint(real(dummy(1,1))))
      Allocate (work(lwork))

!     Compute the singular values and left and right singular vectors
!     of A.

!     The NAG name equivalent of dgesvd is f08kpf
      Call zgesvd('A','S',m,n,a,lda,s,u,ldu,vt,ldvt,work,lwork,rwork,info)

      If (info/=0) Then
        Write (nout,99999) 'Failure in F08KPF/ZGESVD. INFO =', info
99999   Format (1X,A,I4)
        Go To 100
      End If

!     Print the significant singular values of A

      Write (nout,*) 'Singular values of A:'
      Write (nout,99998) s(1:min(m,n))
99998 Format (1X,4(3X,F11.4))
```

```
      If (prerr>0) Then
        Call compute_error_bounds(m,n,s)
      End If

      If (m>n) Then
!       Compute V*Inv(S)*U^T * b to get least-squares solution.
        Call compute_least_squares(m,n,a_copy,m,u,ldu,vt,ldvt,s,b)
      End If

100   Continue

    Contains
      Subroutine compute_least_squares(m,n,a,lda,u,ldu,vt,ldvt,s,b)

!       .. Use Statements ..
        Use nag_library, Only: dznrm2, zgemv
!       .. Implicit None Statement ..
        Implicit None
!       .. Scalar Arguments ..
        Integer, Intent (In)                    :: lda, ldu, ldvt, m, n
!       .. Array Arguments ..
        Complex (Kind=nag_wp), Intent (In)      :: a(lda,n), u(ldu,m), vt(ldvt,n)
        Complex (Kind=nag_wp), Intent (Inout)   :: b(m)
        Real (Kind=nag_wp), Intent (In)         :: s(n)
!       .. Local Scalars ..
        Complex (Kind=nag_wp)                   :: alpha, beta
        Real (Kind=nag_wp)                      :: norm
!       .. Local Arrays ..
        Complex (Kind=nag_wp), Allocatable      :: x(:), y(:)
!       .. Intrinsic Procedures ..
        Intrinsic                               :: allocated, cmplx
!       .. Executable Statements ..
        Continue
        Allocate (x(n),y(n))

!       Compute V*Inv(S)*U^H * b to get least-squares solution.

!       y = U^T b
!       The NAG name equivelent of zgemv is f06saf
        alpha = cmplx(1.0_nag_wp,0.0_nag_wp,kind=nag_wp)
        beta = cmplx(0.0_nag_wp,0.0_nag_wp,kind=nag_wp)
        Call zgemv('C',m,n,alpha,u,ldu,b,1,beta,y,1)

        y(1:n) = y(1:n)/s(1:n)

!       x = V y
        Call zgemv('C',n,n,alpha,vt,ldvt,y,1,beta,x,1)

        Write (nout,*)
        Write (nout,*) 'Least squares solution:'
        Write (nout,99999) x(1:n)

!       Find norm of residual ||b-Ax||.
        alpha = cmplx(-1.0_nag_wp,0.0_nag_wp,kind=nag_wp)
        beta = cmplx(1._nag_wp,0.0_nag_wp,kind=nag_wp)
        Call zgemv('N',m,n,alpha,a,lda,x,1,beta,b,1)

        norm = dznrm2(m,b,1)

        Write (nout,*)
        Write (nout,*) 'Norm of Residual:'
        Write (nout,99998) norm

        If (allocated(x)) Deallocate (x)
        If (allocated(y)) Deallocate (y)

99999   Format (4X,'(',F8.4,',',F8.4,')')
99998   Format (4X,F11.4)

      End Subroutine compute_least_squares
```

```
      Subroutine compute_error_bounds(m,n,s)

!       Error estimates for singular values and vectors is computed
!       and printed here.

!       .. Use Statements ..
        Use nag_library, Only: ddisna, nag_wp, x02ajf
!       .. Implicit None Statement ..
        Implicit None
!       .. Scalar Arguments ..
        Integer, Intent (In)                    :: m, n
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (In)         :: s(n)
!       .. Local Scalars ..
        Real (Kind=nag_wp)                      :: eps, serrbd
        Integer                                 :: i, info
!       .. Local Arrays ..
        Real (Kind=nag_wp), Allocatable         :: rcondu(:), rcondv(:),        &
                                                   uerrbd(:), verrbd(:)
!       .. Executable Statements ..
        Continue
        Allocate (rcondu(n),rcondv(n),uerrbd(n),verrbd(n))

!       Get the machine precision, EPS and compute the approximate
!       error bound for the computed singular values.  Note that for
!       the 2-norm, S(1) = norm(A)

        eps = x02ajf()
        serrbd = eps*s(1)

!       Call DDISNA (F08FLF) to estimate reciprocal condition
!       numbers for the singular vectors

        Call ddisna('Left',m,n,s,rcondu,info)
        Call ddisna('Right',m,n,s,rcondv,info)

!       Compute the error estimates for the singular vectors

        Do i = 1, n
          uerrbd(i) = serrbd/rcondu(i)
          verrbd(i) = serrbd/rcondv(i)
        End Do

!       Print the approximate error bounds for the singular values
!       and vectors

        Write (nout,*)
        Write (nout,*) 'Error estimate for the singular values'
        Write (nout,99999) serrbd
        Write (nout,*)
        Write (nout,*) 'Error estimates for the left singular vectors'
        Write (nout,99999) uerrbd(1:n)
        Write (nout,*)
        Write (nout,*) 'Error estimates for the right singular vectors'
        Write (nout,99999) verrbd(1:n)

99999   Format (4X,1P,6E11.1)

      End Subroutine compute_error_bounds

    End Program f08kpfe
```

9.2 Program Data

```
F08KPF Example Program Data

   6            4                                          : m and n

 ( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
```

```
(-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
(-0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26) : Matrix A(1:m,1:n)

( 1.00, 0.00) ( 1.00, 0.00) ( 1.00, 0.00)
( 1.00, 0.00) ( 1.00, 0.00) ( 1.00, 0.00)               : RHS b(1:n)
```

9.3 Program Results

```
F08KPF Example Program Results

Singular values of A:
        3.9994        3.0003        1.9944        0.9995

Least squares solution:
   ( -0.0719, -0.2761)
   (  0.6796,  0.4326)
   ( -0.3832, -0.5447)
   (  0.0096, -0.3489)

Norm of Residual:
        1.5266
```